Obyčejné diferenciální rovnice s počítačovou podporou - Maple

Petr Kundrát

Ústav matematiky, FSI VUT v Brně

Tento soubor vznikl za účelem ilustrace použití prostředí Maple k řešení a vizualizaci řešení obyčejných diferenciálních rovnic. Od čtenáře se předpokládá elementární orientace v prostředí Maple a základní znalosti z oblasti obyčejných diferenciálních rovnic.

```
> restart; # tento příkaz vyčistí paměť programu Maple
```

1 ODR 1. řádu

```
> restart;
```

V této kapitole se budeme zabývat řešením diferenciálních rovnic prvního řádu. Ukážeme si, jak formulovat diferenciální rovnici, získat obecné řešení, partikulární řešení počátečního problému a na závěr i vykreslení směrového pole ODR1. Nejprve tedy naformulujeme obyčejnou diferenciální rovnici $\frac{d}{dx}\,\mathrm{y}(x) = -\frac{1\,\mathrm{y}(x)}{2}$. Pro efektivnější práci s touto rovnicí si ji uložíme do výrazu MojeODR1:

```
> MojeODR1 := diff(y(x),x) = -1/2*y(x);
```

$$\mathit{MojeODR1} := \frac{d}{dx}\,\mathrm{y}(x) = -\frac{1}{2}\,\mathrm{y}(x)$$

Poznámka. Je důležité, aby neznámá funkce v diferenciální rovnici byla zapisována i s její nezávisle proměnnou (tedy v našem příkladu $\mathrm{y}(x)$). V opačném případě dojde později k chybám, neboť Maple pokládá neznámou bez nezávisle proměnné za parametr. Chybný zápis výše formulované rovnice by byl

> MojeODR := diff(y(x),x) = -1/2*y;

Obecné řešení rovnice MojeODR1 získáme použitím příkazu dsolve. Parametry tohoto příkazu jsou diferenciální rovnice MojeODR1 a neznámá, vzhledem ke které tuto rovnici řešíme, tedy $\mathrm{y}(x)$. Řešení si lze uložit do výrazu ObecneReseni prostřednictvím přiřazení a odkazu na pravou stranu (rhs) posledního výstupu (%).

```
> dsolve(MojeODR1,y(x));
```

$$\mathrm{y}(x) = \_C1\, e^{(-\frac{x}{2})}$$

```
> ObecneReseni:=rhs(%);
```

$$\mathit{ObecneReseni} := \_C1\, e^{(-\frac{x}{2})}$$

Integrační konstantu v řešení ODR označuje Maple jako "*_C1*". Nyní se pokusíme určit partikulární řešení počátečního problému, sestávajícího z diferenciální rovnice MojeODR1 a počáteční podmínky $\mathrm{y}(0) = 3$. Použijeme opět příkaz dsolve s parametry: dsolve({diferenciální rovnice, počáteční podmínka}, hledaná funkce y(x));

```
> dsolve({MojeODR1, y(0)=3}, y(x));
```

$$\mathrm{y}(x) = 3\, e^{(-\frac{x}{2})}$$

Pro vykreslení řešení počátečního problému lze nyní využít jak klasický příkaz plot pro vykreslení pravé strany předchozího výrazu,

```
> plot(rhs(%), x= 0 .. 5);
```



tak i příkaz DEplot z knihovny DEtools. Parametry tohoto p říkazu jsou:

DEplot(diferenciální rovnice, neznámá, rozsah grafu v x-ové ose (příp. i y-ové ose), [seznam počátečních podmínek], další volitelné parametry - viz help)

Pokud chceme použít příkaz z dosud nezahrnuté knihovny (výše by muselo být "with(DEtools);"), můžeme použít následující syntaxi.

```
>  DEtools[DEplot](MojeODR1, y(x), x=0..5, [[y(0) = 3],[y(0) = 4]], colour=grey,
linecolor=[red,blue]);
```



Jak je z obrázku zřejmé, lze do jednoho takového grafu vykreslit i více partikulárních řešení, tedy v našem případě y(0) = 3 a y(0) = 4. Navíc je v grafu šedou barvou vykresleno pole orientovaných úseček (směrové pole), které mají tu vlastnost, že se ve svých středech dotýkají jednotlivých integrálních křivek příslušejících nějaké počáteční podmínce.

Poznámka. Hledáme-li řešení obyčejné diferenciální rovnice, můžeme se programu Maple otázat, o jakou diferenciální rovnici se jedná. K tomu nám poslouží příkaz odeadvisor z knihovny DEtools, který nám zadanou rovnici klasifikuje.

```
> DEtools[odeadvisor](MojeODR1);
```

$$[\_quadrature]$$

O jednotlivých typech rovnic: viz nápověda k příkazu odeadvisor.

```
> ? odeadvisor
```

1.1 Vyzkoušejte si sami:

Nalezněte partikulární řešení počátečních problémů a nakreslete jejich grafy.

a) y'-3y = 8x - sin(x), y(0) = 1.

b) $\frac{d}{dx}\,\mathrm{y}(x) = -4\,x\,\mathrm{y}(x) + 2\,x\,e^{(-x^2)}\,\sqrt{\mathrm{y}(x)}$, y(0) = 2.

```
> restart;
```

2 ODR n-tého řádu

```
> restart;
```

V této části se budeme zabývat řešením diferenciálních rovnic vyššího řádu. Ukážeme si, jak získat obecné řešení, partikulární řešení počátečního problému a na závěr i vykreslení fázového portrétu řešení. Jako ilustrativní příklad si naformulujeme lineární obyčejnou diferenciální rovnici 3. řádu s konstantními koeficienty, kterou si uložíme do výrazu MojeODR3:

```
> MojeODR3 := diff(y(x),x$3)-5*diff(y(x),x$2)+3*diff(y(x),x)
> -3*y(x)=0;
```

$$MojeODR3 := (\frac{d^3}{dx^3}\,\mathrm{y}(x)) - 5\,(\frac{d^2}{dx^2}\,\mathrm{y}(x)) + 3\,(\frac{d}{dx}\,\mathrm{y}(x)) - 3\,\mathrm{y}(x) = 0$$

Obecné řešení rovnice MojeODR získáme použitím příkazu dsolve. Parametry tohoto příkazu jsou diferenciální rovnice MojeODR3 a neznámá, vzhledem ke které tuto rovnici řeším, tedy y(x).

```
> dsolve(MojeODR3,y(x));
```

$$\mathrm{y}(x) = \_C1\,e^{\left(\frac{(\%1+16+5\,(98+18\,\sqrt{17})^{(1/3)})\,x}{3\,(98+18\,\sqrt{17})^{(1/3)}}\right)}$$
$$-\,\_C2\,e^{\left(-\frac{(\%1+16-10\,(98+18\,\sqrt{17})^{(1/3)})\,x}{6\,(98+18\,\sqrt{17})^{(1/3)}}\right)}\,\sin(\frac{(\sqrt{3}\,\%1-16\,\sqrt{3})\,x}{6\,(98+18\,\sqrt{17})^{(1/3)}})$$
$$+\,\_C3\,e^{\left(-\frac{(\%1+16-10\,(98+18\,\sqrt{17})^{(1/3)})\,x}{6\,(98+18\,\sqrt{17})^{(1/3)}}\right)}\,\cos(\frac{(\sqrt{3}\,\%1-16\,\sqrt{3})\,x}{6\,(98+18\,\sqrt{17})^{(1/3)}})$$
$$\%1 := (98+18\,\sqrt{17})^{(2/3)}$$

Pro vyčíslení poněkud nepřehledného, symbolicky počítaného výrazu lze použít příkaz evalf:

```
> evalf(%);
```

$$\mathrm{y}(x) = \_C1\,e^{(4.479815750\,x)} - 1.\,\_C2\,e^{(0.2600921257\,x)}\,\sin(0.7759010800\,x)$$
$$+\,\_C3\,e^{(0.2600921257\,x)}\,\cos(0.7759010800\,x)$$

K vykreslení grafu řešení MojeODR3 s počátečními podmínkami y(0) = 1, y'(0) = 2, y"(0) = 1 využijeme příkaz DEplot. Poznamenejme, že počáteční podmínky je třeba zadat ve formě funkcí, a pomocí operátoru derivace "D". Nelze použít výraz, tedy formu subs(x = 0, diff(y(x), x)).

```
> DEtools[DEplot](MojeODR3, y(x), x = 0..2, [[y(0) = 1,D(y)(0) = 2,(D@@2)(y)(0)
= 1]], linecolor = [blue],stepsize =.05);
```



2.1 Vyzkoušejte si sami:

Nalezněte obecné řešení a řešení počátečních problémů následujících úloh. Klasifikujte řešené rovnice.

a) sin(x)y”(x) + 5cos(x)y’(x)-x = 5y(x).

b) y”’(x) - tg(x)y(x) = cos(x), y(0) = 1, y’(0) = 0, y”(0) = 3.

c) y”(x) - sin(y(x)) = 0, y(1) = 1, y’(1) = 2.

3 Systémy ODR 1. řádu

```
> restart;
```

V této kapitole budeme řešit systémy obyčejných diferenciálních rovnic 1. řádu. Zaměříme se na systém tří rovnic 1. řádu:

```
> MujSystem:=diff(x(t),t)=x(t)-y(t),diff(y(t),t)=z(t)-y(t),diff(z(t),t)=-3*y(t);
```

$$\mathit{MujSystem} := \frac{d}{dt}\,\mathrm{x}(t) = \mathrm{x}(t) - \mathrm{y}(t),\ \frac{d}{dt}\,\mathrm{y}(t) = \mathrm{z}(t) - \mathrm{y}(t),\ \frac{d}{dt}\,\mathrm{z}(t) = -3\,\mathrm{y}(t)$$

Obecné řešení opět získáme pomocí příkazu dsolve.

```
> dsolve({MujSystem},{x(t),y(t),z(t)});
```

$$\{\mathrm{x}(t) = \frac{1}{10}\,\_C2\,e^{(-\frac{t}{2})}\cos(\frac{\sqrt{11}\,t}{2})\,\sqrt{11} + \frac{3}{10}\,\_C2\,e^{(-\frac{t}{2})}\sin(\frac{\sqrt{11}\,t}{2})$$
$$+ \frac{3}{10}\,\_C3\,e^{(-\frac{t}{2})}\cos(\frac{\sqrt{11}\,t}{2}) - \frac{1}{10}\,\_C3\,e^{(-\frac{t}{2})}\sin(\frac{\sqrt{11}\,t}{2})\,\sqrt{11} + e^t\,\_C1, \mathrm{z}(t) = \frac{1}{2}$$
$$e^{(-\frac{t}{2})}$$
$$(\_C2\sin(\frac{\sqrt{11}\,t}{2}) + \_C2\cos(\frac{\sqrt{11}\,t}{2})\,\sqrt{11} + \_C3\cos(\frac{\sqrt{11}\,t}{2}) - \_C3\sin(\frac{\sqrt{11}\,t}{2})\,\sqrt{11}),$$
$$\mathrm{y}(t) = e^{(-\frac{t}{2})}\,(\_C2\sin(\frac{\sqrt{11}\,t}{2}) + \_C3\cos(\frac{\sqrt{11}\,t}{2}))\}$$

Následující grafy nám umožňují zobrazit závislosti mezi jednotlivými proměnnými t, x(t), y(t), z(t).

```
> with(DEtools):
> DEplot3d({MujSystem},[x(t),y(t),z(t)],t=0..10,[[x(0)=0,y(0)=1,z(0)=-9]],
 scene=[t,x(t),y(t)],stepsize=.1, title=`Reseni systemu v souradnicich
 t, x(t), y(t)`,linecolor=t);
```




```
> DEplot3d({MujSystem},[x(t),y(t),z(t)],t=0..10,[[x(0)=0, y(0)=1,z(0)=-9]],
 scene=[t, x(t), z(t)],stepsize=.1, title=`Reseni systemu v souradnicich
 t, x(t), z(t)`,linecolor=t);
```




```
> DEplot3d({MujSystem},[x(t),y(t),z(t)],t=0..10,[[x(0)=0,y(0)=1,z(0)=-9]],
 scene=[t,y(t),z(t)],stepsize=.1, title=`Reseni systemu v souradnicich
 t, y(t), z(t)`,linecolor=t, axes = normal);
```




```
>  DEplot3d({MujSystem},[x(t),y(t),z(t)],t=0..10,[[x(0)=0,
>  y(0)=1,z(0)=-9]],scene=[x(t),y(t),z(t)],stepsize=.1,
>  title=‘Reseni systemu v souradnicich x(t), y(t), z(t)‘,linecolor=t,
>  axes = normal);
```



Pro vykreslení dvourozměrného grafu (ortogonální projekce řešení do roviny dvou řešených proměnných) lze využít příkazu phaseportrait. Níže vidíme řešení soustavy MujSystem v rovině (y,z):

```
>  phaseportrait([MujSystem],[x(t),y(t),z(t)],t=-2.5..9,[[x(0)=0,y(0)=1,
>  z(0)=-9]],stepsize=.05,scene=[y(t),z(t)],linecolor=t);
```



Stejnou situaci uvidíte vhodným natočením grafu DEplot3d v souřadnicích y(t), z(t) (viz výše).